
T E C H N I C A L  N O T E  
Doc No.: TN30-24 
Date: March 22, 2007 
Version: 1.0 
Subject: Simple Shift Register (FIFO) Example 

 

This program example demonstrates a simple FIFO shift register.  The term FIFO stands for "first in 
first out".  This technique is useful when an event needs to be recorded and acted on at some later 
time.  A typical example of this would be for product reject handling.  Where product defects are 
detected at one machine station, the bad product is tracked though the machine as it indexes, and then 
ejected at some later station. 

 

Fig. 1 - Machine line diagram. 

 

The data input is a TRUE/FALSE condition from the inspection station that needs to be shifted through 
the register.  In this example we are going to use five VR variables to store the data.  Pointer variables 
are used to manipulate where the data is loaded and unloaded from the FIFO. 

The program can be tested by using the MotionPerfect Digital I/O Status to force output 8 which 
simulates the inspection station sensor.  The product reject output will be indicated by output 9.  The 
FIFO shift data is displayed in the MotionPerfect Terminal 5 window. 

     Fig. 2 - Digital I/O Status window. 

 

 

Fig. 3 - Program output display. 

 

'============================================================================== 

' Filename: FIFO.BAS 

' Revisions: 

'   21Mar07 - Initial release. (AMS) 

'------------------------------------------------------------------------------ 

' Abstract: 

' This program demonstrates a simple FIFO shift register. 

'============================================================================== 

 

'--Zero VR(0) thru VR(4) used to store the five register FIFO data. 

VR(0)=0 

VR(1)=0 

VR(2)=0 

VR(3)=0 

VR(4)=0 

 

'--Init some variables. 

i_pointer=0 'input data VR pointer 

o_pointer=1 'output data VR pointer 

inspect_i=8 'input point for product inspection station 

reject_o=9 'output point for product reject station 

dwell=500 'shift dwell time 

 

OP(reject_o,OFF) 'force product reject output signal off 

 

'--Main loop. 

WHILE 1=1 

  i_pointer = (i_pointer+1) MOD(5) 'increment FIFO input pointer 

  o_pointer = (o_pointer+1) MOD(5) 'increment FIFO output pointer 

 

  '--Check product inspection input to load FIFO. 

  IF IN(inspect_i)=ON THEN 

    VR(i_pointer)=1 'load 1 for rejected product 

  ELSE 

    VR(i_pointer)=0 'load 0 for good product 

  ENDIF 

 

  '--Check FIFO output pointer to fire product reject output. 

  IF VR(o_pointer)=1 THEN OP(reject_o,ON) 

  WA(dwell) 

  OP(reject_o,OFF) 

 

  GOSUB display_fifo 

  WA(dwell) 

WEND 

 

display_fifo: 

  pa=(i_pointer+4) MOD(5) 

  pb=(i_pointer+3) MOD(5) 

  pc=(i_pointer+2) MOD(5) 

  pd=(i_pointer+1) MOD(5) 

  PRINT #5,VR(i_pointer)[0];VR(pa)[0];VR(pb)[0];VR(pc)[0];VR(pd)[0] 

RETURN 

'============================================================================== 

 


